
T TEST

PURPOSE
Perform a two sample t test.

DESCRIPTION
This tests the hypothesis that two population means are equal. That is:

H0: u1 = u2
Ha: u1 <> u2
Test Statistic (assuming equal population variances):

(EQ 3-55)

where Sp is the pooled standard deviation:

(EQ 3-56)

The degrees of freedom equals n1+n2-2.
Test Statistic (assuming unequal population variances):

(EQ 3-57)

The degrees of freedom equals (var1/n1 + var2/n2)2/denom where denom equals (var1/n1)2/(n1-1) + (var2/n2)2/(n2-1)
and var1 and var2 are the sample variances for the two samples and n1 and n2 are the sample sizes.

Significance level: Typically set to .05
Critical Region:

T < -t(alpha/2,df), T > t(alpha/2,df)
where df is the degrees of freedom. The t values can be computed as:

LET TALPHA = TPPF(.975,DF)
Conclusion: Reject null hypothesis if T in critical region

SYNTAX
T TEST <y1> <y2> <SUBSET/EXCEPT/FOR qualification>
where <y1> is a variable containing the values from the first sample;

 <y2> is a variable containing the values from the second sample;
and where the <SUBSET/EXCEPT/FOR qualification> is optional.

EXAMPLES
T TEST Y1 Y2
T TEST Y1 Y2 SUBSET TAG > 2

NOTE 1
The sample sizes for the two variables do not need to be equal.

NOTE 2
DATAPLOT automatically prints the test statistic for both the equal and unequal population variances assumptions.

NOTE 3
Although DATAPLOT does not treat paired observations as a special case, the test can be computed as follows:

LET D = Y1 - Y2
LET DBAR = MEAN D
LET DSD = STANDARD DEVIATION D

LET N = SIZE D
LET T = DBAR/(DSD/SQRT(N))
LET CRITICAL = TPPF(.975,N-1)

The value of T is then compared against the value of CRITICAL.

NOTE 4
When the normality assumption is suspect, there are several non-parametric alternatives. The Wilcoxon rank sum test (also called the
Mann-Whitney U test) can be used for the unpaired t test. The sign test or the Wilcoxon signed rank test can be used for the paired case.
Although DATAPLOT does not support these tests directly, they are straight forward to implement as macros. This is demonstrated in
program examples 2 through 4 below.

NOTE 5
The various values printed by the F TEST command are saved as parameters

DEFAULT
None

SYNONYMS
None

RELATED COMMANDS
CONFIDENCE LIMITS = Compute the confidence limits for the mean of a sample.
F TEST = Carry out a 2-sample test for the equality of the standard deviations.
CHI-SQUARE TEST = Carry out a 1-sample chi-square test for the standard deviation equal to some

specified value.
BIHISTOGRAM = Generates a bihistogram.
QUANTILE-QUANTILE PLOT = Generate a quantile-quantile plot.
BOX PLOT = Generates a box plot.

REFERENCE
T tests are discussed in most introductory statistics books.

APPLICATIONS
Confirmatory Data Analysis

IMPLEMENTATION DATE
87/4 (the output format was modified 94/2, the automatic saving of parameters was added 94/12)

PROGRAM 1
SKIP 25; READ AUTO83B.DAT Y1 Y2
RETAIN Y2 SUBSET Y2 > -999
T TEST Y1 Y2

The following output is generated.

                       T TEST
                     (2-SAMPLE)
 HYPOTHESIS BEING TESTING--POPULATION MEANS MU1 = MU2

 SAMPLE 1:
    NUMBER OF OBSERVATIONS      =      249
    MEAN                        =    20.14458
    STANDARD DEVIATION          =    6.414700
    STANDARD DEVIATION OF MEAN  =   0.4065151

 SAMPLE 2:
    NUMBER OF OBSERVATIONS      =       79
    MEAN                        =    30.48101
    STANDARD DEVIATION          =    6.107710
    STANDARD DEVIATION OF MEAN  =   0.6871710

 IF     ASSUME SIGMA1 = SIGMA2:
    POOLED STANDARD DEVIATION   =    6.342600
    DIFFERENCE (DEL) IN MEANS   =   -10.33643
    STANDARD DEVIATION OF DEL   =   0.8190135
    T TEST STATISTIC VALUE      =   -12.62059
    DEGREES OF FREEDOM          =    326.0000
    T TEST STATISTIC CDF VALUE  =    0.000000

 IF NOT ASSUME SIGMA1 = SIGMA2:
    STANDARD DEVIATION SAMPLE 1 =    6.414700
    STANDARD DEVIATION SAMPLE 2 =    6.107710
    BARTLETT CDF VALUE          =    0.402799
    DIFFERENCE (DEL) IN MEANS   =   -10.33643
    STANDARD DEVIATION OF DEL   =   0.7984100
    T TEST STATISTIC VALUE      =   -12.94627
    EQUIVALENT DEG. OF FREEDOM  =    136.8750
    T TEST STATISTIC CDF VALUE  =    0.000000

 HYPOTHESIS     ACCEPTANCE INTERVAL    CONCLUSION
 MU1 < MU2         (0.000,0.950)         REJECT
 MU1 = MU2         (0.025,0.975)         REJECT
 MU1 > MU2         (0.050,1.000)         REJECT

 PARAMETER INFINITY  HAS THE VALUE:     0.3402823E+39
 PARAMETER PI        HAS THE VALUE:     0.3141593E+01
 PARAMETER STATVAL   HAS THE VALUE:    -0.1262059E+02
 PARAMETER STATNU    HAS THE VALUE:     0.3260000E+03
 PARAMETER POOLSD    HAS THE VALUE:     0.6342600E+01
 PARAMETER STATCDF   HAS THE VALUE:    -0.3330669E-15
 PARAMETER CUTLOW95  HAS THE VALUE:    -0.1967268E+01
 PARAMETER CUTUPP95  HAS THE VALUE:     0.1967268E+01
 PARAMETER CUTLOW99  HAS THE VALUE:    -0.2590994E+01
 PARAMETER CUTUPP99  HAS THE VALUE:     0.2590995E+01

PROGRAM 2
.  Perform a Wilcoxon rank sum (also called a Mann-Whitney U) non-parametric 2-sample t-test.
.
SKIP 25
READ AUTO83B.DAT X1 X2
RETAIN X2 SUBSET X2 > -999
LET N1 = SIZE X1; LET N2 = SIZE X2; LET N = MIN(N1,N2)
.
LET TAG = 1 FOR I = 1 1 N1; LET TAG2 = 2 FOR I = 1 1 N2
LET X = X1; EXTEND X X2; EXTEND TAG TAG2
.
LET X = SORTC X TAG
LET X = RANK X
LET W1 = SUM X SUBSET TAG = 1
LET W2 = SUM X SUBSET TAG = 2
LET U1 = W1 - N1*(N1+1)/2
LET U2 = W2 - N2*(N2+1)/2
LET U = MIN(U1,U2)
.
FEEDBACK OFF
IF N > 8

LET UU = N1*N2/2
LET SIGMA = SQRT(N1*N2*(N1+N2+1)/12)
LET Z = (U - UU)/SIGMA
LET ALPHA = 0.05
LET ALPHA2 = 1.0 - ALPHA/2
LET CRITICAL = NORPPF(ALPHA2)
PRINT “ “; PRINT “H0: U1 = U2”
PRINT “HA: U1 <> U2”
PRINT “WILCOXON SIGNED RANK U STATISTIC = ^U”
PRINT “NORMAL APPROXIMATION FOR WILCOXON SIGNED RANK U STATISTIC = ^Z”
PRINT “NORMAL CRITICAL VALUE = +/- ^CRITICAL”
LET Z2 = ABS(Z)
IF Z2 <= CRITICAL

PRINT “ACCEPT NULL HYPOTHESIS AT THE ^ALPHA SIGNIFICANCE LEVEL”
END OF IF
IF Z2 > CRITICAL

PRINT “REJECT NULL HYPOTHESIS AT THE ^ALPHA SIGNIFICANCE LEVEL”
END OF IF

END OF IF
IF N <= 8

PRINT “ “; PRINT “H0: U1 = U2”
PRINT “HA: U1 <> U2”
PRINT “WILCOXON SIGNED RANK U STATISTIC = ^U”
PRINT “COMPARE THIS VALUE TO TABLES FOR THE WILCOXON”
PRINT “SIGNED RANK TEST.”

END OF IF

The following output is generated.

H0: U1 = U2
HA: U1 <> U2
WILCOXON SIGNED RANK U STATISTIC = 2521.5
NORMAL APPROXIMATION FOR WILCOXON SIGNED RANK U STATISTIC = -9.95943
NORMAL CRITICAL VALUE = +/- 1.959965
REJECT NULL HYPOTHESIS AT THE 0.05 SIGNIFICANCE LEVEL

PROGRAM 3
. Perform a non-parameteric sign test for a paired sample.
. This test can be easily adapted to a one-sample hypothesis test
. that the mean (or median) is equal to a given value.
. Data from “Probability and Statistics for Engineers and
. Scientists” by Walpole and Myers (example 13.3 on page 483).
.
LET X1 = DATA 4.2 4.7 6.6 7.0 6.7 4.5 5.7 6.0 7.4 4.9 6.1 5.2
LET X2 = DATA 4.1 4.9 6.2 6.9 6.8 4.4 5.7 5.8 6.9 4.7 6.0 4.9
.
. SET D0 TO CONSTANT YOU WANT TO TEST AGAINST.
. THAT IS D0 = 0 TESTS U1 = U2 (OR U1 - U2 = 0)
. WHILE D0 = 5 TESTS U1 - U2 =5.
LET D0 = 0
.
LET DIFF = X1 - X2 - D0
LET N = SIZE DIFF SUBSET DIFF <> 0
LET RPLUS = SIZE DIFF SUBSET DIFF > 0
LET RMINUS = SIZE DIFF SUBSET DIFF < 0
LET R = MIN(RPLUS,RMINUS)
LET P =0.5
.
FEEDBACK OFF
LET ALPHA = 0.05
LET CRITICAL = BINPPF(ALPHA,0.5,N)
CAPTURE SIGN_OUT.DAT
PRINT “ “
PRINT “H0: U1 - U2 = ^D0”
PRINT “HA: U1 - U2 <> ^D0”
PRINT “SIGN STATISTIC = ^R”
PRINT “BINOMIAL CRITICAL VALUE = ^CRITICAL”
IF R >= CRITICAL

PRINT “ACCEPT NULL HYPOTHESIS AT THE ^ALPHA SIGNIFICANCE LEVEL”
END OF IF
IF R < CRITICAL

PRINT “REJECT NULL HYPOTHESIS AT THE ^ALPHA SIGNIFICANCE LEVEL”
END OF IF

The following output is generated.

H0: U1 - U2 = 0
HA: U1 - U2 <> 0
SIGN STATISTIC = 2
BINOMIAL CRITICAL VALUE = 3
REJECT NULL HYPOTHESIS AT THE 0.05 SIGNIFICANCE LEVEL

PROGRAM 4
.  Perform a non-parameteric Wilcoxon signed rank test for a paired  sample. Data from
.  “Probability and Statistics for Engineers and Scientists” by Walpole and Myers (example 13.3
.  on page 483).
.  SET D0 TO CONSTANT YOU WANT TO TEST AGAINST.
.  THAT IS D0 = 0 TESTS U1 = U2 (OR U1 - U2 = 0)
.  WHILE D0 = 5 TESTS U1 - U2 = 5.
.
LET X1 = DATA 4.2 4.7 6.6 7.0 6.7 4.5 5.7 6.0 7.4 4.9 6.1 5.2
LET X2 = DATA 4.1 4.9 6.2 6.9 6.8 4.4 5.7 5.8 6.9 4.7 6.0 4.9
LET D0 = 0
.
LET DIFF = X1 - X2 - D0
RETAIN DIFF SUBSET DIFF <> 0
LET N = SIZE DIFF
LET TAG = 1 FOR I = 1 1 N
LET TAG = 2 SUBSET DIFF < 0
LET ADIFF = ABS(DIFF)
LET R = RANK ADIFF
LET WPLUS = SUM R SUBSET TAG = 1
LET WMINUS = SUM R SUBSET TAG = 2
LET W = MIN(WPLUS,WMINUS)
.
FEEDBACK OFF
IF N > 30

LET UU = N*(N+1)/4
LET SIGMA = SQRT(N*(N+1)*(2*N+1)/24)
LET Z = (W - UU)/SIGMA
LET ALPHA = 0.05
LET ALPHA2 = 1.0 - ALPHA/2
LET CRITICAL = NORPPF(ALPHA2)
PRINT “ “; PRINT “H0: U1 - U2 = ^D0”
PRINT “HA: U1 - U2 <> ^D0”
PRINT “WILCOXON RANKED SIGN STATISTIC = ^W”
PRINT “NORMAL APPROXIMATION FOR WILCOXON RANKED SIGN STATISTIC = ^Z”
PRINT “NORMAL CRITICAL VALUE = +/- ^CRITICAL”
LET Z2 = ABS(Z)
IF Z2 <= CRITICAL

PRINT “ACCEPT NULL HYPOTHESIS AT THE ^ALPHA SIGNIFICANCE LEVEL”
END OF IF
IF Z2 > CRITICAL

PRINT “REJECT NULL HYPOTHESIS AT THE ^ALPHA SIGNIFICANCE LEVEL”
END OF IF

END OF IF
.  WCRIT CONTAINS THE CRITICAL VALUES FOR A TWO-SIDED TEST AND
.  ALPHA = 0.05.  CONSULT TABLES FOR DIFFERENT SIGNIFICANCE LEVEL.
.  N NEEDS TO NE AT LEAST 6 IN ORDER TO OBTAIN A 0.05 SIGNIFICANCE LEVEL.
.
IF N <= 30
  SERIAL READ WCRIT
  0 0 0 0 0 1 2 4 6 8 11 14 17 21 25 30 35 40 46 52 59 66 73
  81 90 98 107 117 127 137
  END OF DATA
  LET CRITICAL = WCRIT(N)
  PRINT “ “; PRINT “H0: U1 - U2 = ^D0”
  PRINT “HA: U1 - U2 <> ^D0”
  PRINT “WILCOXON RANKED SIGN STATISTIC = ^W”

PRINT “CRITICAL VALUE = ^CRITICAL”
IF W > CRITICAL

PRINT “ACCEPT NULL HYPOTHESIS AT THE 0.05 SIGNIFICANCE LEVEL”
END OF IF
IF W <= CRITICAL

PRINT “REJECT NULL HYPOTHESIS AT THE 0.05 SIGNIFICANCE LEVEL”
END OF IF

END OF IF

The following output is generated.

H0: U1 - U2 = 0
HA: U1 - U2 <> 0
WILCOXON RANKED SIGN STATISTIC = 12.5
CRITICAL VALUE = 11
ACCEPT NULL HYPOTHESIS AT THE 0.05 SIGNIFICANCE LEVEL


